

*** indices ***
alpha cue_eye cue_move cue_face, item
alpha cue_confidence cue_nervous cue_calm cue_guilt cue_think, item
alpha cue_content cue_detail, item
alpha cue_pause cue_error cue_certainty cue_pitch, item

*** Figure 1 ***
foreach var of varlist cue_eye cue_move cue_face cue_confidence cue_nervous cue_calm cue_guilt cue_think cue_content cue_detail cue_pause cue_error cue_certainty cue_pitch {
	recode `var' (1/3=0) (4/5=1), gen(bi_`var')
	ci proportions bi_`var' if video==1
	drop bi_`var'
} 

*** TRUTH BIAS ***

* truth bias for same-party candidates
prtest truth if inlist(gender,0,1) & include==1 & video!=0, by(party_match)

* Hypothesis 1: Truth bias: people will be more likely to say politicians are telling the truth
bitest truth==.5 if inlist(gender,0,1) & include==1 & video!=0

* truth bias by gender of candidate
bitest truth==.5 if gender_cand==0 & include==1 & video!=0 //male
bitest truth==.5 if gender_cand==1 & include==1 & video!=0 //female candidate
prtest truth if inlist(gender,0,1) & include==1 & video!=0, by(gender_cand) //compare male to female candidate

* results separated into four categories, by gender of respondent & gender of candidates
bysort gender: bitest truth==.5 if gender_cand==0 & include==1 & video!=0
bysort gender: bitest truth==.5 if gender_cand==1 & include==1 & video!=0


*** MORE ABOUT TRUTH JUDGMENTS ***

** confidence that they have detected truth versus deception 
ttest confidence if include==1 & video!=0, by(truth)

* lie more about facts, self, or opponent
tab truth type if include==1 & video!=0, col chi2



*** ACCURACY OF ASSESSMENTS ***

** politician was telling the truth
tab truth_lie if inlist(gender,0,1) & include==1 & video!=0

* Hypothesis 2: people can recognize whether politicians are telling the truth

bitest response=.5 if inlist(gender,0,1) & include==1 & video!=0


** footnote 8
tab truth_lie gender_cand if inlist(gender,0,1) & include==1 & video!=0, col

** footnote 9
tab response type if inlist(gender,0,1) & include==1 & video!=0, col

** accuracy of majority: footnote 10
bysort video: tab response if inlist(gender,0,1) & include==1 & video!=0

* Hypotheses H3 - H6 *
*** Table 2 ***

logit response verbalC demeanorC nonverbalC paraverbalC c.confidence##truth i.type gender gender_cand party_m discuss interest ease lie_point lie_often if inlist(gender,0,1) & include==1 & video!=0, cluster(subj)

logit response cue_detail cue_content demeanorC nonverbalC paraverbalC c.confidence##truth i.type gender gender_cand party_m discuss interest ease lie_point lie_often if inlist(gender,0,1) & include==1 & video!=0, cluster(subj)

margins truth, at(confidence=(0 25 50 75 100))
marginsplot

